Esimerkki: Tietoliikennekytkin

Tämä Mathematica - notebook sisältää luennolla 2A (20.10.2016) käsitellyn esimerkin laskut.

Esimerkin kuvailu

Tarkastellaan yksinkertaista mallia tietoliikennekytkimelle. Kytkimeen saapuu datapaketteja, kytkin tallentaa ne väliaikaisesti puskuriinsa, ja lähettää ne sitten eteenpäin. Pidämme kirjaaerityisesti puskurissa olevien datapakettien määrästä ajan funktiona, jota mallinnetaan Markov-prosessilla. Kullakin ajanhetkellä (esim. 10^{-6} s pituisilla ajanjaksoilla) kytkin lähettää puskuristaan yhden paketin eteenpäin, paitsi puskurin ollessa tyhjä, jolloin kytkin ei lähetä mitään. Kytkimeen saapuu vielä ennen seuraavaa ajanhetkeä Poisson (λ)-jakautunut määrä datapaketteja, riippumattomasti eri aikaväleillä. Saapuvat datapaketit tallennetaan puskuriin, mikäli puskurissa on tilaa. Puskuriin voi kerrallaan olla tallennettuna enintään M datapakettia. Paketit, joita ei voida tallentaa, hylätään.
Jos X_t on ajanhetkellä t puskurissa olevien datapakettien lukumäärä, on ylläolevilla oletuksilla prosessi $X = (X_t)_{t \in \mathbb{Z}_+}$ Markov ketju äärellisellä tilajoukolla $S = \{0, 1, 2, \ \ldots, M-1, M\}$.

Parametrit λ ja M pitäisi realistisemmissa tietoliikennesovelluksissa selvittää kytkimen ja verkon ominaisuuksien ja käyttötarkoituksen perusteella. Parametri λ kuvaa pakettien tulointensiteettiä tietoliikenneverkossa, ja se liittyy siten verkon kuormitukseen kyseisen kytkimen kohdalla. Parametri M on puskurin muistikapasiteetti, ja siis kytkimen ominaisuus. Tässä esimerkissä oletamme parametreille arvot $M = 8$ (enintään kahdeksan datapakettia puskurissa) ja $\lambda = 0.95$ (odotusarvoisesti 0.95 saapuvaa pakettia per aikayksikkö).

Poisson-jakauman pistemassafunktio

```
In[1]:= PoissonPMF[k_] := Exp[-λ] * λ^k / k!
```

Vaihteen siirtymämatriisi

Alustetaan oikean kokoinen matriisi P ensin pelkillä nollilla

```
In[2]:= P = Table[0, {i, 0, 8}, {j, 0, 8}];
```

Käydään läpi rivit, ja lasketaan siirtymätodennäköisyydet Poisson-jakaumasta

In[3]:=
```
(* Lasketaan ensin siirtymätodennäköisyydet tyhjästä puskurista:
    nämä ovat viimeistä lukuunottamatta suoraan Poisson-jakauman
        pistetodennäköisyyksiä; viimeinen on todennäköisyys, että
  saapuvien pakettien lukumäärä on kahdeksan tai enemmän *)
For[in= 0, in< 8, in++,
  P[[1, in+1]] = PoissonPMF[in];]
P[[1, 8+1]] = 1-Sum [P[[1, y+1]], {y, 0, 8}];
(* Lasketaan sitten siirtymätodennäköisyydet puskurista, jossa
    on buf>0 pakettia: nämä ovat jälleen viimeistä lukuunottamatta
          suoraan Poisson-jakauman pistetodennäköisyyksiä; viimeinen
  on todennäköisyys, että saapuvien pakettien lukumäärä riittää
  täyttämään puskurin ja mahdollisesti enemmänkin *)
For[buf = 1, buf ≤ 8, buf ++,
  For[in= 0, buf-1+in< 8, in++,
    P[[buf+1, buf-1+in+1]] = PoissonPMF[in];
  ];
  P[[buf+1, 8+1]] = 1-Sum [P[[buf+1, y+1]], {y, 0, 8}];
]
```

Näytetään siirtymämatriisi yleisellä λ : n arvolla

In[6]:=
```
P // MatrixForm
```

Out[6]//MatrixForm=

$$
\begin{pmatrix}
e^{-\lambda} & e^{-\lambda}\lambda & \frac{1}{2}e^{-\lambda}\lambda^2 & \frac{1}{6}e^{-\lambda}\lambda^3 & \frac{1}{24}e^{-\lambda}\lambda^4 & \frac{1}{120}e^{-\lambda}\lambda^5 & \frac{1}{720}e^{-\lambda}\lambda^6 & \frac{e^{-\lambda}\lambda^7}{5040} & 1-e^{-\lambda}-e^{-\lambda}\lambda-\frac{1}{2}e^{-\lambda}\lambda^2-\frac{1}{6}e^{-\lambda}\ldots \\
e^{-\lambda} & e^{-\lambda}\lambda & \frac{1}{2}e^{-\lambda}\lambda^2 & \frac{1}{6}e^{-\lambda}\lambda^3 & \frac{1}{24}e^{-\lambda}\lambda^4 & \frac{1}{120}e^{-\lambda}\lambda^5 & \frac{1}{720}e^{-\lambda}\lambda^6 & \frac{e^{-\lambda}\lambda^7}{5040} & 1-e^{-\lambda}-e^{-\lambda}\lambda-\frac{1}{2}e^{-\lambda}\lambda^2-\frac{1}{6}e^{-\lambda}\ldots \\
0 & e^{-\lambda} & e^{-\lambda}\lambda & \frac{1}{2}e^{-\lambda}\lambda^2 & \frac{1}{6}e^{-\lambda}\lambda^3 & \frac{1}{24}e^{-\lambda}\lambda^4 & \frac{1}{120}e^{-\lambda}\lambda^5 & \frac{1}{720}e^{-\lambda}\lambda^6 & 1-e^{-\lambda}-e^{-\lambda}\lambda-\frac{1}{2}e^{-\lambda}\lambda^2-\frac{1}{6}e\ldots \\
0 & 0 & e^{-\lambda} & e^{-\lambda}\lambda & \frac{1}{2}e^{-\lambda}\lambda^2 & \frac{1}{6}e^{-\lambda}\lambda^3 & \frac{1}{24}e^{-\lambda}\lambda^4 & \frac{1}{120}e^{-\lambda}\lambda^5 & 1-e^{-\lambda}-e^{-\lambda}\lambda-\frac{1}{2}e^{-\lambda}\lambda\ldots \\
0 & 0 & 0 & e^{-\lambda} & e^{-\lambda}\lambda & \frac{1}{2}e^{-\lambda}\lambda^2 & \frac{1}{6}e^{-\lambda}\lambda^3 & \frac{1}{24}e^{-\lambda}\lambda^4 & 1-e^{-\lambda}-e^{-\lambda}\lambda-\frac{1}{2}\ldots \\
0 & 0 & 0 & 0 & e^{-\lambda} & e^{-\lambda}\lambda & \frac{1}{2}e^{-\lambda}\lambda^2 & \frac{1}{6}e^{-\lambda}\lambda^3 & 1-e^{-\lambda}-e^{-}\ldots \\
0 & 0 & 0 & 0 & 0 & e^{-\lambda} & e^{-\lambda}\lambda & \frac{1}{2}e^{-\lambda}\lambda^2 & 1-e^{-}\ldots \\
0 & 0 & 0 & 0 & 0 & 0 & e^{-\lambda} & e^{-\lambda}\lambda & 1\ldots \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & e^{-\lambda} &
\end{pmatrix}
$$

Asetetaan λ : n arvo

In[7]:=
```
λ = .95;
```

Näytetään siirtymämatriisi kiinnitetyllä λ : n arvolla

In[8]:=
```
P // MatrixForm
```

Out[8]//MatrixForm=

$$
\begin{pmatrix}
0.386741 & 0.367404 & 0.174517 & 0.0552637 & 0.0131251 & 0.00249377 & 0.000394847 & 0.00005\ldots \\
0.386741 & 0.367404 & 0.174517 & 0.0552637 & 0.0131251 & 0.00249377 & 0.000394847 & 0.00005\ldots \\
0 & 0.386741 & 0.367404 & 0.174517 & 0.0552637 & 0.0131251 & 0.00249377 & 0.00039\ldots \\
0 & 0 & 0.386741 & 0.367404 & 0.174517 & 0.0552637 & 0.0131251 & 0.0024\ldots \\
0 & 0 & 0 & 0.386741 & 0.367404 & 0.174517 & 0.0552637 & 0.0131\ldots \\
0 & 0 & 0 & 0 & 0.386741 & 0.367404 & 0.174517 & 0.0552\ldots \\
0 & 0 & 0 & 0 & 0 & 0.386741 & 0.367404 & 0.174\ldots \\
0 & 0 & 0 & 0 & 0 & 0 & 0.386741 & 0.367\ldots \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0.386\ldots
\end{pmatrix}
$$

Lasketaan tilajakauma hetkellä 1 lähtien tyhjästä puskurista

In[9]:=
```
({{1, 0, 0, 0, 0, 0, 0, 0, 0}}.P) // MatrixForm
```

Out[9]//MatrixForm=
```
(0.386741 0.367404 0.174517 0.0552637 0.0131251 0.00249377 0.000394847 0.00005
```

Lasketaan tilajakauma hetkellä 2 lähtien tyhjästä puskurista

In[10]:=
```
({{1, 0, 0, 0, 0, 0, 0, 0, 0}}.P.P) // MatrixForm
```

Out[10]//MatrixForm=
```
(0.291659 0.344569 0.217102 0.0975131 0.0349738 0.0105848 0.00278466 0.0006485
```

Stationaarinen jakauma

Puskurin tilaa kuvaava Markov-ketju on yhtenäinen, koska esimerkiksi siirtymät 0→8→7→6→5→4→3→2→1→0 tapahtuvat positiivisillä todennäköisyyksillä. Siten sillä on Lauseen 3.8 mukaan yksikäsitteinen stationaarinen jakauma. Tämä voidaan ratkaista siirtymämatriisin ominaisarvoa 1 vastaavana vasempana ominaisvektorina, jonka komponenttien summa pitää muistaa normittaa ykköseksi (jotta se olisi todennäköisyysjakauma).

Matriisin P ominaisarvot

In[11]:=
```
Eigenvalues[P]
```

Out[11]= $\{1., 0.929695, 0.744848, 0.501684, 0.269838, 0.103573, 0.0213233, 0.000866028, -4.54311\times10^{-13}\}$

Stationaarinen jakauma π on ominaisarvoa 1 vastaava vasen ominaisvektori todennäköisyysjakaumaksi normalisoituna (ja koska vasen ominaisvektori on transpoosin oikea ominaisvektori, voidaan Mathematicaa pyytää laskemaan se komennolla “Eigenvectors[Transpose[P]]”)

In[12]:=
```
Π = Eigenvectors[Transpose[P], 1] /
      Apply[Plus, Flatten[Eigenvectors[Transpose[P], 1]]];
Π // MatrixForm
```

Out[13]//MatrixForm=
```
(0.0874736 0.138708 0.143786 0.133127 0.120554 0.108895 0.0983592 0.0888454 0.
```

Stationaarisen jakauman tulkintaa

Puskurin tilaa kuvaava Markov-ketju on myös jaksoton, koska vaikkapa mistä tahansa tilasta voidaan yhdellä aika-askeleella päästä itseensä. Lauseen 3.11 perusteella yhtenäisen jaksottoman Markov ketjun hetkittäiset tilajakaumat μ_t suppenevat kun $t \to \infty$ kohti stationaarista jakaumaa π, eli ketju asettuu pitkällä aikavälillä stationaarisen jakauman kuvaamaan tilastolliseen tasapainoon.

Stationaarisessa jakaumassa (johon ketju pitkällä aikavälillä asettuu) puskuri on annetulla ajanhetkellä tyhjänä noin 8.7 % todennäköisyydellä, täynnä noin 8.0 % todennäköisyydellä, ja tyypillisimmin puskurissa on kaksi pakettia, noin 14.4 % todennäköisyydellä.

Pitkän aikavälin kustannuskertymiä

Puskurin käyttöaste

Käyttöastetta kuvaava kustannusfunktio $c : S \to \mathbb{R}$ on yksinkertaisesti $c(x) = x$, missä $x \in S = \{0, 1, 2, 3, 4, 5, 6, 7, 8\}$ on puskurissa olevien pakettien lukumäärä. Funktio tulkitaan tilajoukon S indeksoimana pystyvektorina.

In[14]:=
```
c = Transpose[{{0, 1, 2, 3, 4, 5, 6, 7, 8}}];
c // MatrixForm
```

Out[15]//MatrixForm=

$$\begin{pmatrix} 0 \\ 1 \\ 2 \\ 3 \\ 4 \\ 5 \\ 6 \\ 7 \\ 8 \end{pmatrix}$$

Pitkän aikavälin kustannusvauhti lasketaan stationaarisen jakauman ja kustannusfunktion pistetulona

In[16]:=
```
Π.c
```

Out[16]= {{3.70644}}

Tämä on toki sama asia kuin summa vektorien komponenttien yli

In[17]:=
```
Sum [Π[[1, i]] * c[[i, 1]], {i, 1, 9}]
```

Out[17]= 3.70644

Hylättyjen pakettien odotettu määrä per aikayksikkö

Hylättyjen pakettien odotettua määrää kuvaava kustannusfunktio $c : S \to \mathbb{R}$ saa tilassa $x \in S$ arvon, joka on odotusarvo sille, kuinka paljon Poisson-jakautunut saapuvien pakettien lukumäärä ylittää puskurissa sillä hetkellä käytettävissä olevan kapasiteetin $8 - \max\ \{0, x - 1\}$ (tässä huomioidaan, että ajanhetken aluksi on jo lähetetty puskurista eteenpäin yksi paketti). Funktio tulkitaan jälleen tilajoukon S indeksoimana pystyvektorina, ja se lasketaan seuraavasti.

In[18]:=
```
c = Transpose[{Table[Sum [PoissonPMF[in] * Max[0, in - (8 - Max[0, (x - 1)])],
        {in, 8 - Max[0, x - 1], Infinity}], {x, 0, 8}]}];
c // MatrixForm
```

Out[19]//MatrixForm=

$$\begin{pmatrix} 8.17764\times10^{-7} \\ 8.17764\times10^{-7} \\ 7.92264\times10^{-6} \\ 0.000068614 \\ 0.000524153 \\ 0.00347347 \\ 0.0195479 \\ 0.090886 \\ 0.336741 \end{pmatrix}$$

In[20]:=
```
Π.c
```

Out[20]= {{0.0374736}}

Tämä on jälleen toki sama asia kuin summa vektorien komponenttien yli

In[21]:=
```
Sum [Π[[1, i]] * c[[i, 1]], {i, 1, 9}]
```

Out[21]= 0.0374736

Kulkuaikoja

Odotetut ajat puskurin ensimmäiseen tyhjenemiseen

Oletetaan, että puskurissa on aluksi $X_0 = x$ dakettia. Tarkastellaan satunnaista hetkeä $T_{\{0\}} = \min\ \{t \geq 0 \mid X_t = 0\}$, jolloin puskuri on ensimmäistä kertaa tyhjillään. Lasketaan, miten odotettu kulkuaika $k(x) = \mathrm{E}_x[T_{\{0\}}]$ riippuu alkutilasta x.

Muodostetaan tuntemattomista odotetuista kulkuajoista pystyvektori $(k(x))_{x \in \{0,1,\ldots,8\}}$.

In[22]:=
```
kvec = Transpose[{Map[k[#] &, Range[0, 8]]}];
kvec // MatrixForm
```

Out[23]//MatrixForm=

$$\begin{pmatrix} k[0] \\ k[1] \\ k[2] \\ k[3] \\ k[4] \\ k[5] \\ k[6] \\ k[7] \\ k[8] \end{pmatrix}$$

Käytetään ensin ilmeistä havaintoa $k(0) = 0$.

In[24]:=
```
k[0] = 0
kvec // MatrixForm
```

Out[24]= 0

Out[25]//MatrixForm=

$$\begin{pmatrix} 0 \\ k[1] \\ k[2] \\ k[3] \\ k[4] \\ k[5] \\ k[6] \\ k[7] \\ k[8] \end{pmatrix}$$

Muodostetaan myös vektori, jonka kaikki komponentit ovat ykkösiä.

In[26]:=
```
onevec = Transpose[{Table[1, {x, 0, 8}]}];
onevec // MatrixForm
```

Out[27]//MatrixForm=

$$\begin{pmatrix} 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \end{pmatrix}$$

Odotetut kulkuajat saadaan sitten vaatimalla allaolevan vektorin komponentit (tilaa $x = 0$ vastaavaa komponenttia lukuunottamatta) nolliksi.

In[28]:=
```
((P - IdentityMatrix[9]).kvec + onevec) // MatrixForm
```

Out[28]//MatrixForm=

```
1. + 0.367404 k[1] + 0.174517 k[2] + 0.0552637 k[3] + 0.0131251 k[4] + 0.00249377 k[5] + 0.00
1. - 0.632596 k[1] + 0.174517 k[2] + 0.0552637 k[3] + 0.0131251 k[4] + 0.00249377 k[5] + 0.00
1 + 0.386741 k[1] - 0.632596 k[2] + 0.174517 k[3] + 0.0552637 k[4] + 0.0131251 k[5] + 0.002
1 + 0.386741 k[2] - 0.632596 k[3] + 0.174517 k[4] + 0.0552637 k[5] + 0.0131251 k
1 + 0.386741 k[3] - 0.632596 k[4] + 0.174517 k[5] + 0.0552637 k[6] + 0.
1 + 0.386741 k[4] - 0.632596 k[5] + 0.174517 k[6] + 0.0552637
1 + 0.386741 k[5] - 0.632596 k[6] + 0.174517 k[7] + 0.
1 + 0.386741 k[6] - 0.632596 k[7] + 0.245855
1 + 0.386741 k[7] - 0.386741 k[8]
```

Ratkaistaan tämä lineaarinen yhtälöryhmä.

In[29]:=
```
Solve[Table[((P - IdentityMatrix[9]).kvec + onevec)[[j]] == 0, {j, 2, 9}]]
```

Out[29]=
```
{{k[1] → 11.432, k[2] → 21.9466, k[3] → 31.4455, k[4] → 39.8199,
  k[5] → 46.9495, k[6] → 52.7009, k[7] → 56.9304, k[8] → 59.5161}}
```

Vektorimuodossa ratkaisu on seuraava :

In[30]:=
```
kvecsol = (kvec /. %[[1]]);
kvecsol // MatrixForm
```

Out[31]//MatrixForm=

$$\begin{pmatrix} 0 \\ 11.432 \\ 21.9466 \\ 31.4455 \\ 39.8199 \\ 46.9495 \\ 52.7009 \\ 56.9304 \\ 59.5161 \end{pmatrix}$$

Odotetut ajat puskurin ensimmäiseen täyttymiseen

Oletetaan jälleen, että puskurissa on aluksi $X_0 = x$ dakettia. Tarkastellaan satunnaista hetkeä $T_{\{8\}} = \min\ \{t \geq 0 \mid X_t = 8\}$, jolloin puskuri on ensimmäistä kertaa täynnä. Lasketaan, miten odotettu kulkuaika $k(x) = \mathrm{E}_x[T_{\{0\}}]$ riippuu alkutilasta x.

Muodostetaan tuntemattomista odotetuista kulkuajoista pystyvektori $(k(x))_{x \in \{0,1,\ldots,8\}}$ ja käytetään ilmeistä havaintoa $k(8) = 0$.

In[32]:=
```
k8vec = Transpose[{Map[k8[#] &, Range[0, 8]]}];
k8[8] = 0;
k8vec // MatrixForm
```

Out[34]//MatrixForm=

$$\begin{pmatrix} k8[0] \\ k8[1] \\ k8[2] \\ k8[3] \\ k8[4] \\ k8[5] \\ k8[6] \\ k8[7] \\ 0 \end{pmatrix}$$

Odotetut kulkuajat saadaan sitten vaatimalla allaolevan vektorin komponentit (tilaa $x = 8$ vastaavaa komponenttia lukuunottamatta) nolliksi.

In[35]:=
```
((P - IdentityMatrix[9]).k8vec + onevec) // MatrixForm
```

Out[35]//MatrixForm=
```
1. - 0.613259 k8[0] + 0.367404 k8[1] + 0.174517 k8[2] + 0.0552637 k8[3] + 0.0131251 k8[4] + 0
1. + 0.386741 k8[0] - 0.632596 k8[1] + 0.174517 k8[2] + 0.0552637 k8[3] + 0.0131251 k8[4] + 0
          1. + 0.386741 k8[1] - 0.632596 k8[2] + 0.174517 k8[3] + 0.0552637 k8[4] + 0.01312
                   1. + 0.386741 k8[2] - 0.632596 k8[3] + 0.174517 k8[4] + 0.0552637 k8[5]
                            1. + 0.386741 k8[3] - 0.632596 k8[4] + 0.174517 k8[5] + 0.0552
                                    1. + 0.386741 k8[4] - 0.632596 k8[5] + 0.174517 k8[6
                                            1. + 0.386741 k8[5] - 0.632596 k8[6] + 0.174
                                                    1. + 0.386741 k8[6] - 0.632596 k8[7
                                                            1. + 0.386741 k8[7]
```

Ratkaistaan tämä lineaarinen yhtälöryhmä.

In[36]:=
```
Solve[Table[((P - IdentityMatrix[9]).k8vec + onevec)[[j]] == 0, {j, 1, 8}]]
```

Out[36]=
```
{{k8[0] → 82.9352, k8[1] → 82.9352, k8[2] → 80.7936, k8[3] → 76.2811,
   k8[4] → 69.1433, k8[5] → 59.0998, k8[6] → 45.8871, k8[7] → 29.6341}}
```

Vektorimuodossa ratkaisu on seuraava :

In[37]:=
```
k8vecsol = (k8vec /. %[[1]]);
k8vecsol // MatrixForm
```

Out[38]//MatrixForm=

$$\begin{pmatrix} 82.9352 \\ 82.9352 \\ 80.7936 \\ 76.2811 \\ 69.1433 \\ 59.0998 \\ 45.8871 \\ 29.6341 \\ 0 \end{pmatrix}$$